
Explicit code mentions arguments:

 Expression (⌈/N)-⌊/N←3 1 4 1 5

 Tradfn ∇ R←Range Y …

 Dfn {(⌈/⍵)-(⌊/⍵)}

Tacit code implies arguments:

 Tacit ⌈/-⌊/

What is the essence of 
tacit programming? Find out here!

f/   f¨   ∘.g   f\   A∘g   f⌸   f⍠B

APL old-timers don’t want you to know this one fact:

Operator application is actually just tacit programming!

Shocking Revelation:
You’ve been unknowingly using 
tacit constructs all along!

f⍥g   f⍤g   f⍨   fgh   f∘g

Discover the mind-blowing secret:

Function composition is actually just plumbing!

Function composition

 Arguments in operands

 Memorable (like ≠⊆⊢ and +⌿÷≢)

 Adjacency (like ×- and ∨/⍷)

 Brevity (like F⍥⎕C)

 DRY (Don’t Repeat Yourself; like ≡⍥⍴)

 Just a general feeling of superiority and awesomeness

Experts rave about tacit!

The shape of an outer product ⍺ ∘.f ⍵ is
(⍴⍺) , (⍴⍵)

We can write this as
⍺ ,⍥⍴ ⍵

“pre-process both”

Over

Location of ⍺th 1 in each element of ⍵ is
⍺ ⊃¨ ⍸¨ ⍵

We can write this as
⍺ ⊃∘⍸¨ ⍵

“pre-process right”

Beside

Any-presence of ⍺ in ⍵ is 
∨/ ⍺ ⍷ ⍵

We can write this as
⍺ ∨/⍤⍷ ⍵

“post-processing result”

Atop

A multiplication table of N is
(⍳⍵) ∘.× (⍳⍵)

We can write this as
∘.×⍨ ⍳⍵

“selfie”

Commute



Over Beside Atop Commute

pre-process
both

pre-process
right

post-process
result

selfie

Tacit Techniques
Function Compositions

1

1 0

4 10 0

¯10 4 0 0

1.  'Hello' {(⎕C ⍺)≡(⎕C ⍵)} 'HELLO'

2.  'ab' 'cd' {∧/⍺∊⍵}¨ 'abba' 'dad'

3.  64 1000 0 {⍺*÷⍵} 3

4.  10 4 1 0 {⍺××⍵} ¯3 2 0 ¯1

Tasks: Tacify!

Also a type of composition

Sequence of functions in isolation:

 Parenthesised:
 (+⌿÷≢) 3 1 4 1 5

 Assigned
 Avg←+⌿÷≢
 Avg 3 1 4 1 5

Train Introduction

(f Y) + (h Y)  →  (f + h) Y

Discussion: Writing Trains

(f Y) + (h Y)  →  (f + h) Y

(f Y) + ( Y)  →  (f + ⊢) Y

Discussion: Writing Trains

(f Y) + (h Y)  →  (f + h) Y

(f Y) + ( Y)  →  (f + ⊢) Y

(X f Y) + (X h Y)  →  X (f + h) Y

Discussion: Writing Trains

(f Y) + (h Y)  →  (f + h) Y

(f Y) + ( Y)  →  (f + ⊢) Y

(X f Y) + (X h Y)  →  X (f + h) Y

(X   ) + (X h Y)  →  X (⊣ + h) Y

Discussion: Writing Trains



Tacit Techniques17

(f Y) + (h Y)  →  (f + h) Y

(f Y) + ( Y)  →  (f + ⊢) Y

(X f Y) + (X h Y)  →  X (f + h) Y

(X   ) + (X h Y)  →  X (⊣ + h) Y

Discussion: Writing Trains

(f Y) g (h Y)  →  (f g h) Y

(f Y) g ( Y)  →  (f g ⊢) Y

  g (h Y)  →  ( g h) Y

(X f Y) g (X h Y)  →  X (f g h) Y

(X   ) g (X h Y)  →  X (⊣ g h) Y

Discussion: Writing Trains

(f Y) g (h Y)  →  (f g h) Y

(f Y) g ( Y)  →  (f g ⊢) Y

  g (h Y)  →  ( g h) Y

(X f Y) g (X h Y)  →  X (f g h) Y

(X   ) g (X h Y)  →  X (⊣ g h) Y

      g (X h Y)  →  X ( g h) Y

Discussion: Writing Trains

(f Y) g (h Y)  →  (f g h) Y

(f Y) g ( Y)  →  (f g ⊢) Y

  g (h Y)  →  ( g h) Y

(X f Y) g (X h Y)  →  X (f g h) Y

(X   ) g (X h Y)  →  X (⊣ g h) Y

      g (X h Y)  →  X ( g h) Y

Discussion: Writing Trains

(f Y) g (h Y)  →  (f g h) Y

(f Y) g ( Y)  →  (f g ⊢) Y

  f (g Y)  →  ( f g) Y

(X f Y) g (X h Y)  →  X (f g h) Y

(X   ) g (X h Y)  →  X (⊣ g h) Y

      f (X g Y)  →  X ( f g) Y

Discussion: Writing Trains

Some parts can actually be arrays:
 A g h is {A} g h
Two sub-types of trains:
 Fork: f g h and A g h
 Atop: f g
Longer trains:
 (d e f g h) is (d e (f g h))
 (e f g h) is (e (f g h))

Train Details



Monadic
(f Y) g (h Y) ↔ (f g h) Y

(f Y) g ( Y) ↔ (f g ⊢) Y

  g (h Y) ↔ ( g h) Y

Dyadic
(X f Y) g (X h Y) ↔ X (f g h) Y

(X   ) g (X h Y) ↔ X (⊣ g h) Y

      g (X h Y) ↔ X ( g h) Y

Tacit Techniques
Trains

{∧/ (⌈\⍵) = ⍵}  1 3 5 6 7 
1
      {∧/ (⌈\⍵) = (⊢⍵)}  1 3 5 6 7 
1
      {∧/ (⌈\ = ⊢) ⍵}  1 3 5 6 7 
1
      {(∧/ ⌈\ = ⊢) ⍵}  1 3 5 6 7 
1
      (∧/ ⌈\ = ⊢)  1 3 5 6 7 
1

Example: Writing a Train

4 14 2 16

3 4 6

1 2 5 10

4

1.  {2×⍵} 2 7 1 8

2.  3 1 4 {(⍺∪⍵)~(⍺∩⍵)} 1 6 1

3.  {∪⍵∨⍳⍵} 10

4.  2 {(⍺⊃⍒⍵)⊃⍵} 3 1 4 1 5

Tasks: Tacify!

Amazing tool: ]box on

┌─┬─────────────────┐
│|│┌─┬─┬───────────┐│
│ ││⊢│÷│┌─────┬─┬─┐││
│ ││ │ ││┌─┬─┐│⍣│≡│││
│ ││ │ │││+│/││ │ │││
│ ││ │ ││└─┴─┘│ │ │││
│ ││ │ │└─────┴─┴─┘││
│ │└─┴─┴───────────┘│
└─┴─────────────────┘

]box on |⊢÷+/⍣≡

-t=box

┌─┬─────────────────┐
│|│┌─┬─┬───────────┐│
│ ││⊢│÷│┌─────┬─┬─┐││
│ ││ │ ││┌─┬─┐│⍣│≡│││
│ ││ │ │││+│/││ │ │││
│ ││ │ ││└─┴─┘│ │ │││
│ ││ │ │└─────┴─┴─┘││
│ │└─┴─┴───────────┘│
└─┴─────────────────┘

]box on –t=… |⊢÷+/⍣≡

-t=box -t=tree

┌─┬─────────────────┐
│|│┌─┬─┬───────────┐│
│ ││⊢│÷│┌─────┬─┬─┐││
│ ││ │ ││┌─┬─┐│⍣│≡│││
│ ││ │ │││+│/││ │ │││
│ ││ │ ││└─┴─┘│ │ │││
│ ││ │ │└─────┴─┴─┘││
│ │└─┴─┴───────────┘│
└─┴─────────────────┘

┌─┴─┐   .
 | ┌─┼─┐ .
   ⊢ ÷ ⍣ .
      ┌┴┐.
      / ≡.
    ┌─┘  .
    +    .

]box on –t=… |⊢÷+/⍣≡

-t=box -t=tree -t=parens

┌─┬─────────────────┐
│|│┌─┬─┬───────────┐│
│ ││⊢│÷│┌─────┬─┬─┐││
│ ││ │ ││┌─┬─┐│⍣│≡│││
│ ││ │ │││+│/││ │ │││
│ ││ │ ││└─┴─┘│ │ │││
│ ││ │ │└─────┴─┴─┘││
│ │└─┴─┴───────────┘│
└─┴─────────────────┘

┌─┴─┐   .
 | ┌─┼─┐ .
   ⊢ ÷ ⍣ .
      ┌┴┐.
      / ≡.
    ┌─┘  .
    +    .

|(⊢÷((+/)⍣≡))

]box on –t=… |⊢÷+/⍣≡

3

 ab  de  fgh

2.25 2.75

2.8

1.  {(⊖⍵)⊥(⊖⍵)} 1 1 1 0 1 1 0 0

2.  ',;' {(~⍵∊⍺)⊆⍵} 'ab,de;fgh'

nums←3 1 4 1 5

3.  4 {(⍺+⌿⍵)÷⍺} nums

4.  {(+⌿⍵)÷≢⍵} nums

5.  Bonus task: Combine 3 & 4 into an ambivalent function. 

Tasks: Tacify!



Tacit Techniques32

 Arguments in operands

 Monadic functions

 Recursion, Assignment, Dotting

 Selection (but 20.0…)

4 aspects of tacit that could ruin 
your life – #4 will blow your mind!

10 {x←⍺ ⋄ ⌽@{x<⍵}⍵} 1 2 13 14 5 16 7 18
1 2 18 16 5 14 7 13
      10 {      ⌽@(⍺<⊢)⍵} 1 2 13 14 5 16 7 18
1 2 18 16 5 14 7 13

      2 {(⌽⍤⍉⍣⍺)⍵} 3 3⍴⍳9
9 8 7
6 5 4
3 2 1

The ultimate paradox revealed:
Arguments in operands

{⌽+⌿↑⌽¨⍵}

⌽⍤(+⌿)⍤↑⍤(⌽¨)

⌽(+⌿(↑⌽¨))

⌽+⌿⍤↑⍤(⌽¨)

⌽(+⌿⍤↑⌽¨)

Inappropriate tacit discovered: 
Lots of monadic functions

 Assignment

 Namespace “dotting”

 Recursion

{n⍪⍉↑⍺⍎¨n←⍺.⎕NL¯2}
   ┘ │
     │
     ┘

          ┐
{1≥⍵:1 ⋄ (⍵-1)+⍥∇(⍵-2)}

3 things tacit code just cannot do



Tacit Techniques36

{(3>⍵)⌿⍵} 3 1 4 1 5
4 5
       (3∘>⌿⊢)  3 1 4 1 5
SYNTAX ERROR
      (3∘>⌿ ⊢) 3 1 4 1 5
SYNTAX ERROR
     (3∘> ⊢⍤⌿ ⊢)3 1 4 1 5
4 5
        3∘>⍛⌿   3 1 4 1 5

The ugly truth about selection,
and what we plan to do about it

'aeiou' {⍵[⍺⍋⍵]} 'hello world'
eoohll wrld
      'aeiou' {⍵⌷⍨⍺⍋⍵} 'hello world'
LENGTH ERROR
      'aeiou' {⍵⌷⍨⊂⍺⍋⍵} 'hello world'
eoohll wrld
      'aeiou' (⊂⍤⍋⌷⊢) 'hello world'
eoohll wrld
      'aeiou' (⍋⊇⊢) 'hello world'

The ugly truth about selection,
and what we plan to do about it

1. Monadic ××⌊⍤|

2. Dyadic ⌊∘≢↑⊢

3. Monadic ≡⍥(⎕C~∘' ')

4. Monadic +⌿⊢>+⌿÷≢

Bonus tasks: 

5. Monadic ⌽≡⊢⊢≡⌽

Tasks: Convert tacit to dfn

Amazing tool: tacit.help

P ⍋ s
P ⍒ s
P ⍳ s
P ∪ s
s ∊ P
s ~ P
s ∩ P

P∘⍋ s
P∘⍒ s
P∘⍳ s
P∘∪ s

(∊∘P) s
(~∘P) s
(∩∘P) s

7 hash tables that won’t go away

s←'Hello, World!'
    AVi←⎕AV∘⍳ ⋄ {}AVi s
    cmpx '⎕AV⍳s' 'AVi s'
⎕AV⍳s → 6.0E¯7 |   0% ⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕
AVi s → 3.7E¯7 | -39% ⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕        
    AVg←⎕AV∘⍋ ⋄ {}AVg s
    cmpx '⎕AV⍋s' 'AVg s'
⎕AV⍋s → 2.3E¯6 |   0% ⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕
AVg s → 5.5E¯7 | -77% ⎕⎕⎕⎕⎕               

7 hash tables that won’t go away

s←'Hello, World!'

    cmpx '⎕AV⍳s' '⎕AV∘⍳ s'
⎕AV⍳s   → 6.5E¯7 |  0% ⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕ 
⎕AV∘⍳ s → 7.0E¯7 | +7% ⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕

    cmpx '⎕AV⍋s' '⎕AV∘⍋ s'
⎕AV⍋s   → 2.0E¯6 |     0% ⎕                   
⎕AV∘⍋ s → 6.7E¯5 | +3242% ⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕

7 hash tables that won’t go away

⍪e
┌─────────────┐
│a (∧/¨∊¨) b  │
├─────────────┤
│a {∧/¨⍺∊¨⍵} b│
├─────────────┤
│a ∧/⍤∊¨ b    │
├─────────────┤
│a {∧/⍺∊⍵}¨ b │
└─────────────┘

a←(2e4⍴¨'ab' 'cd')[?5e4⍴2]
b←     'aba' 'cad'[?5e4⍴2]

∇ hwm←mem expr
  {}⎕WA
  {}0(2000⌶)14
  hwm←2000⌶14
  {}⍎expr
  hwm-⍨←2000⌶14
∇

Showdown: Memory vs CPU

(⍕⍪,mem¨) e
 a (∧/¨∊¨) b    125835728
 a {∧/¨⍺∊¨⍵} b  125835792
 a ∧/⍤∊¨ b        4194304
 a {∧/⍺∊⍵}¨ b     4194304
      (⍕⍪,1∘cmpx¨) e
 a (∧/¨∊¨) b    0.45675
 a {∧/¨⍺∊¨⍵} b  0.4145 
 a ∧/⍤∊¨ b      0.4245 
 a {∧/⍺∊⍵}¨ b   0.42275

Showdown: Memory vs CPU

a←2↑¨a(⍕⍪,mem¨) e
 a (∧/¨∊¨) b          0
 a {∧/¨⍺∊¨⍵} b        0
 a ∧/⍤∊¨ b      4194304
 a {∧/⍺∊⍵}¨ b   4194304
      (⍕⍪,1∘cmpx¨) e
 a (∧/¨∊¨) b    0.0080625   
 a {∧/¨⍺∊¨⍵} b  0.0078359375
 a ∧/⍤∊¨ b      0.019703125 
 a {∧/⍺∊⍵}¨ b   0.021765625 

Showdown: Memory vs CPU

Function composition:
⍥ Pre-process both
∘ Pre-process right
⍤ Post-process
⍨ Selfie

Operators: long left scope
Trains: odd-even from right
Tools: ]box on -t=…
 tacit.help

Watch out for these:
Arguments in operands
Lots of monadic functions
Just don’t try:
 Assignment
 Namespace “dotting”
 Recursion

Selection issues:
 Compress: ⊢⍤⌿
 Index: ⊂⍤… ⌷ …

Check out this cool summary
— you won’t regret it!

